2017

GAMS Handout 2

Ethan Yang
yey217@lehigh.edu

Using GAMS for Reservoir Operation

Y. C. Ethan Yang
Department of Civil and Environmental Engineering
Lehigh University
Bethlehem, PA USA

YOU WILL LEARN

- Quick review of GAMS example
- Single reservoir operation – theory
- Single reservoir operation – GAMS example
- Single reservoir operation – Homework
- Next step – River basin management

QUICK REVIEW OF GAMS EXAMPLE

- Example setting
  Let's say you are a farmer that wants to decide what crop you want to grow in your farm to **maximize** your profit. You have two different chooses: **wheat** and **cotton**. They require different amount of resources (**water** and **land**) and give you different net profit. All data you need are given in the following tables.



Net profit	USD
wheat	2
cotton	3

Resources need	water	land
wheat	2	3
cotton	4	2

Total available resources

water	20
land	25

- Review the basic components in GAMS: **Set, Parameter, Table, Scalar, Variable, Equation, Model and Solver**.
- This example is one of the fundamental questions in water resource management but only represents a small scale issue. Next, let's enlarge the spatial scale to river network and consider a reservoir operation question that tries to optimize the water needs from downstream.

SINGLE RESERVOIR OPERATION – THEORY

- There are three major different purposes for reservoir operation: water supply, flooding control, hydropower generation



Q is the inflow, **R** is the outflow, **S** is the live storage and **K** is the capacity. **t** represent the time period used for modeling.

- **Water supply**

Water supply optimization models may be formulated to maximize the water that can be supplied given a reservoir volume, giving an indication of the size for reservoir needed for a specified demand. Alternatively, the model may be formulated to prescribe the operations that would allow the water to be supplied to a specified demand as reliably as possible, which a given reservoir volume. The third formulation below shows a model formulated to maximize the economic value of water supply releases. In all cases <u>**the decision variables are the releases**</u>.

  - *Type 1 – maximize total water supply*

$$
\begin{aligned}
&\text{Maximize} \quad WS \\
&\text{s.t.} \\
&S_{t+1} = S_t + Q_t - R_t - L_t \quad t = 1,\ldots,T;\ T+1 = 1 \\
&\boxed{R_t} = \beta_k \cdot WS + SP_t \quad t = 1,\ldots,T \\
&\qquad k = t - 12 \cdot Int\left[\frac{t-12}{12}\right] \\
&S_t \le K \quad t = 1,\ldots,T \\
&\beta_k \cdot WS \le WD_k \quad t = 1,\ldots,T
\end{aligned}
$$

WS is the annual total water supply, WD_k is the water demand in month k, β_k is fixed ratio of month demand to annual demand, SP_t is the spill in time period t

- *Type 2 – maximize reliability*

$$
\begin{aligned}
&\text{Maximize} \quad WS/WD + \lambda \cdot \min_{k=1,2..12}\left(\beta_k \cdot WS / WD_k\right) \\
&\text{s.t.} \\
&S_{t+1} = S_t + Q_t - R_t - L_t \quad t = 1,\ldots,T;\ T+1 = 1 \\
&\boxed{R_t} = \beta_k \cdot WS + SP_t \quad t = 1,\ldots,T \\
&\qquad k = t - 12 \cdot Int\left[\frac{t-12}{12}\right] \\
&S_t \le K \quad t = 1,\ldots,T \\
&\beta_k \cdot WS \le WD_k \quad t = 1,\ldots,T
\end{aligned}
$$

WD is the annual water demand and λ is the weight to the second reliability term

- *Type 3 – real time operation*

$$
\begin{aligned}
&\text{Maximize} \quad \sum_t \left(WS_t / WD_t\right) + \lambda \cdot \min_t\left(WS_t / WD_t\right) \\
&\text{st:} \\
&S_{t+1} = S_t + Q_t - R_t - L_t \quad t = 1,\ldots,T;\ T+1 = 1 \\
&\boxed{R_t} = WS_t + SP_t \quad t = 1,\ldots,T \\
&S_t \le K \quad t = 1,\ldots,T \\
&WS_t \le WD_t \quad t = 1,\ldots,T
\end{aligned}
$$

WD_t is the water demand in time period t, t= 1, 2, .. T. (This value is only available for a real time operation)

- *Type 4 – Maximize water use profit – extended objective function*
  (Will see more in the example)

$$
\begin{aligned}
&\text{Maximize} \quad P(WS_t \mid t = 1,2,.T) \\
&\text{st:} \\
&S_{t+1} = S_t + Q_t - R_t - L_t \quad t = 1,\ldots,T;\ T+1 = 1 \\
&\boxed{R_t} = WS_t + SP_t \\
&S_t \le K \quad t = 1,\ldots,T
\end{aligned}
$$

Develop a profit function for water supply WS_t for one or more particular uses such as irrigation, industry, hydropower generation, etc.

- **Flood control**

Reservoirs with flooding control purpose are either single purpose or coupled with other objectives, in which case storage for flood control (essentially empty space in the reservoir) is a hard constraint on storage levels in the optimization model. The following example shows the reservoir used for water supply but with a constraint for flooding control. Again, **<u>the decision variables are the releases</u>**.

$$\text{Maximize} \quad WS$$

$$\text{s.t.}$$

$$S_{t+1} = S_t + Q_t - R_t - L_t \quad t = 1,\dots,T;\ T+1 = 1$$

$$\boxed{R_t} = \beta_k \cdot WS + SP_t \qquad t = 1,\dots,T$$

$$k = t - 12 \cdot Int\left[\frac{t-12}{12}\right]$$

$$S_t \le K - v_k \qquad t = 1,\dots,T$$

$$\beta_k \cdot WS \le WD_k \qquad t = 1,\dots,T$$

ν is the storage reserved for flooding control in month k

- **Hydropower generation**

This optimization model will attempt to release water to produce energy that matches a schedule of energy demand (Type 1). Alternative formulations for hydropower production might maximize energy production (regardless of demand) or maximize economic production by taking into account the price of energy and it is changes at different times of year (Type 2). In the hydropower generation case, **<u>both the release and the head (storage) are decision variables</u>**.

  - Reservoir with power plant



- Head-storage relationship

$$\overline{H}_t = \frac{H(S_t) + H(S_{t+1})}{2}$$





- *Type1 - Minimizing difference between energy produced and demanded*

$$\text{Minimize} \quad \sum_{t=1}^{T} (E_t - D_t)^2$$

subject to

$$S_{t+1} = S_t + Q_t - R_t - L_t$$
$$S_t \le K \qquad t = 1, \ldots, T$$
$$H_t = H_o + cS_t$$
$$\boxed{\overline{H}_t} = \frac{(H_t - H_o) + (H_{t+1} - H_o)}{2}$$
$$q_t = \boxed{R_t} / time_t \le q_{\max}$$
$$P_t = (9.81)\varepsilon \overline{H}_t q_t \le P_{\max}$$
$$E_t = P_t * time(t) / 3600$$

E is energy generation, *D* is energy demand, *H* is head, *q* is release rate, *P* is power generation, ε is efficiency.

- *Type2 – Maximizing economic production of hydropower generation*

$$\text{Maximize} \quad \sum_{t=1}^{T} E_t \times price_t$$

subject to

$$S_{t+1} = S_t + Q_t - R_t - L_t$$
$$S_t \le K \qquad t = 1, \ldots, T$$
$$H_t = H_o + cS_t$$
$$\boxed{\overline{H}_t} = \frac{(H_t - H_o) + (H_{t+1} - H_o)}{2}$$
$$q_t = \boxed{R_t} / time_t \le q_{\max}$$
$$P_t = (9.81)\varepsilon \overline{H}_t q_t \le P_{\max}$$
$$E_t = P_t * time(t) / 3600$$

price is the energy price at time t.

SINGLE RESERVOIR OPERATION – GAMS EXAMPLE

- GAMS example of reservoir operation for maximize water use profit (adapted and modified from example in Chapter 4, Loucks et al., 2005 and McKinney and Savitsky, 2006)

Let's say this time you are the manager of the upstream reservoir and your job is to release water **every month** for downstream water uses. There are **three firms** downstream and each one of them has their own **profit function**. All data you need are given in the following tables.

- Monthly inflow data are given (implies perfect knowledge of inflows)
- Assuming the total water uses from all three firms is equal to the monthly release and no cap on monthly release.
- In this case, we temporarily ignore the upstream-downstream relationship between firms.

- Schematic



- Profit function for firms



$NB_n(X_n) = aX_n + bX_n^2$

	Firm 1	Firm 2	Firm 3
a	280	389	482
b	-0.8	-0.5	-0.3

Water use, X, in unit of cubic meter (m^3) of water and Net Benefit (NB) in units of dollar

- Inflow

m^3	t_1	t_2	t_3	t_4	t_5	t_6	t_7	t_8	t_9	t_{10}	t_{11}	t_{12}
Q	426	399	523	875	2026	3626	2841	1469	821	600	458	413

- Reservoir information
  Capacity: 9,500 m^3; dead storage: 5,500 m^3; initial storage: 8,000 m^3

- Let's see the GAMS code

```
SETS
i demand sites /i1, i2, i3/
t time / t1*t12/
;

PARAMETERS
a(i) coefficient of benefit function (ax+bx^2)/
i1 280
i2 389
i3 482/

b(i) coefficient of benefit function (ax+bx^2)/
i1 -0.8
i2 -0.5
i3 -0.3/

Q(t) inflow /
t1 426
t2 399
t3 523
t4 875
t5 2026
t6 3626
t7 2841
t8 1469
t9 821
t10 600
t11 458
t12 413/
;

SCALAR
 K reservoir capacity /9500/
 S_min reservoir dead storage /5500/
```

```
 beg_S reservoir initial storage /8000/
;

POSITIVE VARIABLES
S(t)          reservoir storage at time t,
R(t)          reservoir release at time t,
W(i,t)        water use from site i at time t
;

VARIABLES
Obj           objective value;

EQUATIONS
Objective     Objective function,
Balance(t)    Water balance,
Watercap(t)   Water use cap
;

Objective..
Obj =E= SUM((i,t),a(i)*W(i,t)+b(i)*W(i,t)**2);

Balance(t)..
S(t) =E= beg_S $(ORD(t) EQ 1) + S(t-1) $ (ORD(t) GT 1) + Q(t) - R(t);

Watercap(t)..
R(t) =E= SUM(i,W(i,t)) ;

S.UP(t) = K;
S.LO(t) = S_min;

MODEL Reservoir / ALL /;
SOLVE Reservoir USING NLP MAXIMIZING Obj;
```

SINGLE RESERVOIR OPERATION –HOMEWORK

- **Dry year inflow**
  Assuming the weather forecast indicates next year will be a dry year with inflow as following, rerun the model and compare the differences of 1) monthly storage, 2) water use at demand sites and 3) profit at demand sites.

```
Q(t) inflow for dry year/
t1 311
t2 299
t3 403
t4 655
t5 1734
```

```
t6 2988
t7 2451
t8 1137
t9 698
t10 498
t11 358
t12 303/
;
```

- **Environmental flow at downstream**
  As we discuss in the very first class, the water uses are not limited to human uses only. Assuming a new environmental requirement states that the streamflow downstream of the three firms should be greater than 150 m^3 per month, modify the code to incorporate this new policy.

  Use the normal year inflow and compare the difference of 1) monthly storage, 2) water use in demand sites and 3) profit in demand sites with the case without environmental flow setting. In your homework, you need to explain "**why**" and "**how**" you modified the code.

NEXT STEP – RIVER BASIN MANAGEMENT

- Some issue to think about before we move to the next step
  - Why doesn't the model use all water in January?
  - What if we want to consider the upstream-downstream relationship?
  - What if we want to run the model for multiple years?